In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Theta"];
```

tex

We start by loading the package \verb$KnotTheory`$ --- it is only needed because it has many specific knots pre-defined:

pdf

In[∘]:=
```
<< KnotTheory`
```

pdf

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

tex

Next we quietly define the commands \verb$Rot$, used to compute rotation numbers, and \verb$PolyPlot$, used to plot polynomials as bar codes and as hexagonal QR codes. Neither is a part of the core of the computation of $\Theta$, so neither is shown; yet we do show some usage examples.

pdf

In[∘]:=
```
(* Rot suppressed *)
```

In[∘]:=
```
Rot[pd_PD] := Module[{n, xs, x, rots, Xp, Xm, front = {1}, k},
   n = Length@pd; rots = Table[0, {2 n}];
   xs = Cases[pd, x_X :> { Xp[x[[4]], x[[1]]]  PositiveQ@x
                           Xm[x[[2]], x[[1]]]  True      ];
   For[k = 1, k ≤ 2 n, ++k,
    If[FreeQ[front, -k],
     front = Flatten@Replace[front, k → (xs /. {
            Xp[k, l_] | Xm[l_, k] :> {l + 1, k + 1, -l},
            Xp[l_, k] | Xm[k, l_] :> (++rots[[l]]; {-l, k + 1, l + 1}),
            _Xp | _Xm :> {}
           }), {1}],
     Cases[front, k | -k] /. {k, -k} :> --rots[[k]];
    ]
   ];
   {xs /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}}, rots} ];
Rot[K_] := Rot[PD[K]];
```

pdf

In[∘]:=
```
Rot[Mirror@Knot[3, 1]]
```

Out[∘]=
pdf

```
{{{1, 1, 4}, {1, 3, 6}, {1, 5, 2}}, {0, 0, 0, -1, 0, 0}}
```

tex

We urge the reader to compare the above output with the knot diagram in Section~\ref{ssec:OldFormulas}.

pdf

In[ ]:=
```
(* PolyPlot suppressed *)
```

In[◦]:=
```
PolyPlot1[Δ_] := Module[{crs, m, maxc, minc, s, rect},
  rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
  If[Expand[Δ] === 0, Graphics[],
   m = Max[-Exponent[Δ, T, Min], Exponent[Δ, T, Max]];
   crs = CoefficientRules[T^m Δ, {T}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[crs /. ({x_} → c_) :> {
       Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
       Tooltip[Polygon[({x + m - 1/2, 0} + #) & /@ rect], c T^(x-m)]
      }, AspectRatio → Min[1/5, 1/Sqrt[m + 1]],
    ImagePadding → None, PlotRangePadding → None]
  ]];
Options[PolyPlot2] = {Labeled → False};
PolyPlot2[θ_, OptionsPattern[]] := Module[{crs, m1, m2, maxc, minc, s, hex, p},
  If[Expand[θ] === 0, Graphics[{White, Disk[]}],
   hex = Table[{Cos[α], Sin[α]} / Cos[2 π/12] / 2, {α, 2 π/12, 2 π, 2 π/6}];
   m1 = Max[-Exponent[θ, T1, Min], Exponent[θ, T1, Max]];
   m2 = Max[-Exponent[θ, T2, Min], Exponent[θ, T2, Max]];
   crs = CoefficientRules[T1^m1 T2^m2 θ, {T1, T2}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/Sqrt[2]]},*)
     crs /. ({x1_, x2_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        p = {{1, -1/2}, {0, Sqrt[3]/2}}.{x1 - m1, x2 - m2};
        Tooltip[Polygon[(p + #) & /@ hex], c T1^(x1-m1) T2^(x2-m2)],
        If[Not@OptionValue[Labeled], {}, {Black, Text[Style[c T1^(x1-m1) T2^(x2-m2), 30], p]}]
       }
     }, ImagePadding → None, PlotRangePadding → None]
  ]];
PolyPlot[{Δ_, θ_}, opts___Rule] := GraphicsColumn[
  {PolyPlot1[Δ], PolyPlot2[θ, FilterRules[{opts}, Options[PolyPlot2]]]},
  Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
  FilterRules[{opts}, Options[GraphicsColumn]]
 ];
```

In[•]:= 
```
PPDemo = PolyPlot2[2 + T1 - T1 T2 + T2 - T1^-1 + T1^-1 T2^-1 - T2^-1, Labeled → True]
```

Out[•]=




In[•]:= 
```
1 / 2 + T1 - T1 T2 + T2 - T1^-1 + T1^-1 T2^-1 - T2^-1 // TeXForm
```

Out[•]//TeXForm=

```
-T_2 T_1+T_1+T_2-\frac{1}{T_2}+\frac{1}{T_2 T_1}-\frac{1}{T_1}+\frac{1}{2}
```

pdf

In[•]:= 
```
PolyPlot[{T - 1 + T^-1, T1 + 2 T2 - 4 T1^-1 T2^-1}, ImageSize → Tiny]
```

Out[•]=
pdf



tex

We urge the reader to reflect on how the ``QR Code'' part of the above picture corresponds to the 2-variable polynomial $T_1+2T_2-4T_1^{-1}T_2^{-1}$.

Next, we decree that $T_3=T_1T_2$ and define the three ``Feynman Diagram'' polynomials $R_{11}$, $R_{12}$, and $\Gamma_1$:

pdf

In[•]:= 
```
T3 = T1 T2;
```

In[∘]:=
```
CF[ℰ_] := Module[{vs = Union@Cases[ℰ, g__, ∞], ps, c},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ Factor[c] (Times @@ vs^ps)]];
CF[s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj - (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
      g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj + ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
         (T3^s - 1) g3ji (1 - T2^s g1ii - (T1^s - 1) (T2^s + 1) g1ji + (T2^s - 2) g2jj + g2ij)) / (T2^s - 1)) /. s → 1]
```

Out[∘]=

$$\frac{1}{2}+T_2\, g_{1,i,i}\, g_{2,j,i}+\frac{(-1+T_1)\, T_2^2\, g_{1,j,i}\, g_{2,j,i}}{-1+T_2}-g_{1,i,i}\, g_{2,j,j}-\frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{2,j,j}}{-1+T_2}-g_{3,i,i}+(1-T_2)\, g_{2,j,i}\, g_{3,i,i}+2\, g_{2,j,j}\, g_{3,i,i}+\frac{(-1+T_1 T_2)\, g_{3,j,i}}{-1+T_2}-\frac{T_2\, (-1+T_1 T_2)\, g_{1,i,i}\, g_{3,j,i}}{-1+T_2}-\frac{(-1+T_1)\, (1+T_2)\, (-1+T_1 T_2)\, g_{1,j,i}\, g_{3,j,i}}{-1+T_2}+\frac{(-1+T_1 T_2)\, g_{2,i,j}\, g_{3,j,i}}{-1+T_2}+(-1+T_1 T_2)\, g_{2,j,i}\, g_{3,j,i}+\frac{(-2+T_2)\, (-1+T_1 T_2)\, g_{2,j,j}\, g_{3,j,i}}{-1+T_2}+g_{1,i,i}\, g_{3,j,j}+\frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{3,j,j}}{-1+T_2}-g_{2,i,i}\, g_{3,j,j}-T_2\, g_{2,j,i}\, g_{3,j,j}$$

pdf

In[∘]:=

$$R_{11}[\{1, i\_, j\_\}] = \frac{1}{2}+T_2\, g_{1,i,i}\, g_{2,j,i}+\frac{(-1+T_1)\, T_2^2\, g_{1,j,i}\, g_{2,j,i}}{-1+T_2}-g_{1,i,i}\, g_{2,j,j}-\frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{2,j,j}}{-1+T_2}-g_{3,i,i}+(1-T_2)\, g_{2,j,i}\, g_{3,i,i}+2\, g_{2,j,j}\, g_{3,i,i}+\frac{(-1+T_1 T_2)\, g_{3,j,i}}{-1+T_2}-\frac{T_2\, (-1+T_1 T_2)\, g_{1,i,i}\, g_{3,j,i}}{-1+T_2}-\frac{(-1+T_1)\, (1+T_2)\, (-1+T_1 T_2)\, g_{1,j,i}\, g_{3,j,i}}{-1+T_2}+\frac{(-1+T_1 T_2)\, g_{2,i,j}\, g_{3,j,i}}{-1+T_2}+(-1+T_1 T_2)\, g_{2,j,i}\, g_{3,j,i}+\frac{(-2+T_2)\, (-1+T_1 T_2)\, g_{2,j,j}\, g_{3,j,i}}{-1+T_2}+g_{1,i,i}\, g_{3,j,j}+\frac{(-1+T_1)\, T_2\, g_{1,j,i}\, g_{3,j,j}}{-1+T_2}-g_{2,i,i}\, g_{3,j,j}-T_2\, g_{2,j,i}\, g_{3,j,j};$$

In[∘]:=
```
CF[s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj - (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
      g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj + ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
         (T3^s - 1) g3ji (1 - T2^s g1ii - (T1^s - 1) (T2^s + 1) g1ji + (T2^s - 2) g2jj + g2ij)) / (T2^s - 1)) /. s → -1]
```

Out[∘]=

$$-\frac{1}{2}-\frac{g_{1,i,i}\, g_{2,j,i}}{T_2}-\frac{(-1+T_1)\, g_{1,j,i}\, g_{2,j,i}}{T_1\, (-1+T_2)\, T_2}+g_{1,i,i}\, g_{2,j,j}+\frac{(-1+T_1)\, g_{1,j,i}\, g_{2,j,j}}{T_1\, (-1+T_2)}+g_{3,i,i}-\frac{(-1+T_2)\, g_{2,j,i}\, g_{3,i,i}}{T_2}-2\, g_{2,j,j}\, g_{3,i,i}-\frac{(-1+T_1 T_2)\, g_{3,j,i}}{T_1\, (-1+T_2)}+\frac{(-1+T_1 T_2)\, g_{1,i,i}\, g_{3,j,i}}{T_1\, (-1+T_2)\, T_2}-\frac{(-1+T_1)\, (1+T_2)\, (-1+T_1 T_2)\, g_{1,j,i}\, g_{3,j,i}}{T_1^2\, (-1+T_2)\, T_2}-\frac{(-1+T_1 T_2)\, g_{2,i,j}\, g_{3,j,i}}{T_1\, (-1+T_2)}+\frac{(-1+T_1 T_2)\, g_{2,j,i}\, g_{3,j,i}}{T_1\, T_2}+\frac{(-1+2\, T_2)\, (-1+T_1 T_2)\, g_{2,j,j}\, g_{3,j,i}}{T_1\, (-1+T_2)\, T_2}-g_{1,i,i}\, g_{3,j,j}-\frac{(-1+T_1)\, g_{1,j,i}\, g_{3,j,j}}{T_1\, (-1+T_2)}+g_{2,i,i}\, g_{3,j,j}+\frac{g_{2,j,i}\, g_{3,j,j}}{T_2}$$

pdf

In[∘]:=
$$R_{11}[\{-1, i\_, j\_\}] = -\frac{1}{2} - \frac{g_{1,i,i}\, g_{2,j,i}}{T_2} - \frac{(-1+T_1)\, g_{1,j,i}\, g_{2,j,i}}{T_1 (-1+T_2) T_2} + g_{1,i,i}\, g_{2,j,j} + \frac{(-1+T_1)\, g_{1,j,i}\, g_{2,j,j}}{T_1 (-1+T_2)} + g_{3,i,i} - \frac{(-1+T_2)\, g_{2,j,i}\, g_{3,i,i}}{T_2} - 2\, g_{2,j,j}\, g_{3,i,i} - \frac{(-1+T_1 T_2)\, g_{3,j,i}}{T_1 (-1+T_2)} + \frac{(-1+T_1 T_2)\, g_{1,i,i}\, g_{3,j,i}}{T_1 (-1+T_2) T_2} - \frac{(-1+T_1)(1+T_2)(-1+T_1 T_2)\, g_{1,j,i}\, g_{3,j,i}}{T_1^2 (-1+T_2) T_2} - \frac{(-1+T_1 T_2)\, g_{2,i,j}\, g_{3,j,i}}{T_1 (-1+T_2)} + \frac{(-1+T_1 T_2)\, g_{2,j,i}\, g_{3,j,i}}{T_1 T_2} + \frac{(-1+2 T_2)(-1+T_1 T_2)\, g_{2,j,j}\, g_{3,j,i}}{T_1 (-1+T_2) T_2} - g_{1,i,i}\, g_{3,j,j} - \frac{(-1+T_1)\, g_{1,j,i}\, g_{3,j,j}}{T_1 (-1+T_2)} + g_{2,i,i}\, g_{3,j,j} + \frac{g_{2,j,i}\, g_{3,j,j}}{T_2};$$

In[∘]:=
$$\mathrm{CF}\Big[s1 \left(T_1^{s0}-1\right) \left(T_2^{s1}-1\right)^{-1} \left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \text{ /. } \{s0 \to 1, s1 \to 1\}\Big]$$

Out[∘]=
$$\frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{-1+T_2} - \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{-1+T_2} - \frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{-1+T_2} + \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{-1+T_2}$$

pdf

In[∘]:=
$$R_{12}[\{1, i0\_, j0\_\}, \{1, i1\_, j1\_\}] = \frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{-1+T_2} - \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{-1+T_2} - \frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{-1+T_2} + \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{-1+T_2};$$

In[∘]:=
$$\mathrm{CF}\Big[s1 \left(T_1^{s0}-1\right) \left(T_2^{s1}-1\right)^{-1} \left(T_3^{s1}-1\right) g_{1,j1,i0}\, g_{3,j0,i1} \left(\left(T_2^{s0} g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0} g_{2,j1,i0} - g_{2,j1,j0}\right)\right) \text{ /. } \{s0 \to 1, s1 \to -1\}\Big]$$

Out[∘]=
$$-\frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} + \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} + \frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} - \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_1 (-1+T_2)}$$

pdf

In[∘]:=
$$R_{12}[\{1, i0\_, j0\_\}, \{-1, i1\_, j1\_\}] = -\frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,i0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} + \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,i1,j0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} + \frac{(-1+T_1)\, T_2\, (-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,i0}\, g_{3,j0,i1}}{T_1 (-1+T_2)} - \frac{(-1+T_1)(-1+T_1 T_2)\, g_{1,j1,i0}\, g_{2,j1,j0}\, g_{3,j0,i1}}{T_1 (-1+T_2)};$$

In[∘]:=

$$\mathrm{CF}\Big[s1\,\big(T_1^{s0}-1\big)\,\big(T_2^{s1}-1\big)^{-1}\,\big(T_3^{s1}-1\big)\,g_{1,j1,i0}\,g_{3,j0,i1}\,\Big(\big(T_2^{s0}\,g_{2,i1,i0}-g_{2,i1,j0}\big)-\big(T_2^{s0}\,g_{2,j1,i0}-g_{2,j1,j0}\big)\Big)\ /.\ \{s0\to -1,\ s1\to 1\}\Big]$$

Out[∘]=

$$-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,i0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)\,T_2}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,j0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,i0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)\,T_2}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,j0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)}$$

pdf

In[∘]:=

$$R_{12}[\{-1,\ i0\_,\ j0\_\},\ \{1,\ i1\_,\ j1\_\}]=-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,i0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)\,T_2}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,j0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,i0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)\,T_2}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,j0}\,g_{3,j0,i1}}{T_1\,(-1+T_2)};$$

In[∘]:=

$$\mathrm{CF}\Big[s1\,\big(T_1^{s0}-1\big)\,\big(T_2^{s1}-1\big)^{-1}\,\big(T_3^{s1}-1\big)\,g_{1,j1,i0}\,g_{3,j0,i1}\,\Big(\big(T_2^{s0}\,g_{2,i1,i0}-g_{2,i1,j0}\big)-\big(T_2^{s0}\,g_{2,j1,i0}-g_{2,j1,j0}\big)\Big)\ /.\ \{s0\to -1,\ s1\to -1\}\Big]$$

Out[∘]=

$$\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,i0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)\,T_2}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,j0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,i0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)\,T_2}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,j0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)}$$

pdf

In[∘]:=

$$R_{12}[\{-1,\ i0\_,\ j0\_\},\ \{-1,\ i1\_,\ j1\_\}]=\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,i0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)\,T_2}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,i1,j0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)}-\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,i0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)\,T_2}+\frac{(-1+T_1)\,(-1+T_1 T_2)\,g_{1,j1,i0}\,g_{2,j1,j0}\,g_{3,j0,i1}}{T_1^2\,(-1+T_2)};$$

pdf

In[∘]:=

$$\Gamma_1[\varphi\_,\ k\_]=\varphi\,g_{3kk}-\varphi/2;$$

tex

Next comes the main program computing $\Theta$. Fortunately, it matches perfectly with the mathematical description in Section~\ref{sec:Formulas}. In line 01 we let \verb$Cs$ be the list of crossings in an input knot $K$, and $\varphi$ the list of its rotation numbers, using the external program \verb$Rot$ which we have already mentions. We also let $n$ be the length of \verb$Cs$, namely, the number of crossings in $K$. In line 02 we let the starting value of $A$ be the identity matrix, and then in line 03, for each crossing in \verb$Cs$ we add to $A$ a $2\times 2$ block, in rows $i$ and $j$ and columns $i+1$ and $j+1$, as explain in Equation~\eqref{eq:A}. In line 04 we compute the normalized Alexander polynomial $\Delta$ as in~\eqref{eq:Delta}. In line 05 we let $G$ be the inverse of $A$. In line 06 we declare what it means to evaluate, \verb$ev$, a formula $\calE$ that may contain symbols of the form $g_{\nu\alpha\beta}$: each such symbol is to be replaced by the entry in position $\alpha,\-

beta$ of $G$, but with $T$ replaced with $T_\nu$. In line 07 we start computing $\theta$ by computing the first summand in~\eqref{eq:Main}, which in itself, is a sum over the crossings of the knot. In line 08 we add to $\theta$ the double sum corresponding to the second term in~\eqref{eq:Main}, and in line 09, we add the third summand of~\eqref{eq:Main}. Finally, line 10 outputs a pair: $\Delta$, and the re-normalized version of $\theta$.

pdf

In[•]:=
```
Θ[K_] := Θ[K] = Module[{Cs, φ, n, A, Δ, G, ev, θ},
  (* 01 *) {Cs, φ} = Rot[K]; n = Length[Cs];
  (* 02 *) A = IdentityMatrix[2 n + 1];
  (* 03 *) Cases[Cs, {s_, i_, j_} ⧴ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1 ; 0  -1 ))];
  (* 04 *) Δ = T^((-Total[φ] - Total[Cs⟦All,1⟧])/2) Det[A];
  (* 05 *) G = Inverse[A];
  (* 06 *) ev[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} ⧴ (G⟦α, β⟧ /. T → T_ν)];
  (* 07 *) θ = ev[Sum_{k=1}^{n} R_11[Cs⟦k⟧]];
  (* 08 *) θ += ev[Sum_{k1=1}^{n} Sum_{k2=1}^{n} R_12[Cs⟦k1⟧, Cs⟦k2⟧]];
  (* 09 *) θ += ev[Sum_{k=1}^{2 n} Γ_1[φ⟦k⟧, k]];
  (* 10 *)
  Factor@{Δ, (T_1 - 1) (T_2 - 1) (1 - T_1^-1 T_2^-1) (Δ /. T → T_1) (Δ /. T → T_2) (Δ /. T → T_3) θ}
];
```

In[•]:=
```
PolyPlot2[(T_1 - 1) (T_2 - 1) (1 - T_1^-1 T_2^-1)]
```

Out[•]=



tex

On to examples! Starting with the trefoil knot.

pdf

In[∘]:=
```
Expand[Θ[Knot[3, 1]]]
PolyPlot[Θ[Knot[3, 1]], ImageSize → Tiny]
```

pdf

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=
pdf

$$\left\{-1+\frac{1}{T}+T,\right.$$
$$-\frac{1}{T_1^3}-\frac{2}{T_1^2}+\frac{3}{T_1}-3T_1+2T_1^2+T_1^3-\frac{1}{T_2^3}+\frac{1}{T_1^3T_2^3}-\frac{2}{T_1^2T_2^3}+\frac{2}{T_1T_2^3}-\frac{2}{T_2^2}-\frac{2}{T_1^3T_2^2}+\frac{2}{T_1^2T_2^2}+\frac{2T_1}{T_2^2}+\frac{3}{T_2}+\frac{2}{T_1^3T_2}-$$
$$\left.\frac{3}{T_1T_2}-\frac{2T_1^2}{T_2}-3T_2+\frac{2T_2}{T_1^2}+3T_1T_2-2T_1^3T_2+2T_2^2-\frac{2T_2^2}{T_1}-2T_1^2T_2^2+2T_1^3T_2^2+T_2^3-2T_1T_2^3+2T_1^2T_2^3-T_1^3T_2^3\right\}$$

Out[∘]=
pdf





tex

\parpic[r]{\parbox{42mm}{
\includegraphics[width=20mm]{figs/K11n34.png}\hfill\includegraphics[width=20mm]{figs/K11n42.p-ng}
}}
Next are the Conway knot $11_{n34}$ and the Kinoshita-Terasaka knot $11_{n42}$. The two are mutants and famously hard to separate: they both have $\Delta=1$ (as evidenced by their one-bar bar codes below), and they have the same HOMFLY-PT polynomial and Khovanov homology. Yet their $\theta$ invariants are different. Note that the genus of the Conway knot is 3, while the genus of the Kinoshita-Terasaka knot is 2. This agrees with the apparent higher complexity of the QR code of the Conway polynomial, and with the observations in Section~\ref{sec:SandM}.

In[•]:=
```
Expand[Θ[Knot["K11n34"]]]
```

Out[•]=

$$\Big\{1,\ \frac{2}{T_1^6}+\frac{6}{T_1^5}-\frac{6}{T_1^4}-\frac{2}{T_1^3}+\frac{2}{T_1^2}-\frac{18}{T_1}+18\,T_1-2\,T_1^2+2\,T_1^3+6\,T_1^4-6\,T_1^5-2\,T_1^6+\frac{1}{T_1^5\,T_2^7}-\frac{3}{T_1^4\,T_2^7}+\frac{3}{T_1^3\,T_2^7}-\frac{1}{T_1^2\,T_2^7}+$$
$$\frac{2}{T_2^6}-\frac{2}{T_1^6\,T_2^6}+\frac{3}{T_1^5\,T_2^6}-\frac{3}{T_1\,T_2^6}+\frac{6}{T_2^5}+\frac{1}{T_1^7\,T_2^5}+\frac{3}{T_1^6\,T_2^5}-\frac{6}{T_1^5\,T_2^5}+\frac{1}{T_1^4\,T_2^5}-\frac{3}{T_1^3\,T_2^5}+\frac{3}{T_1^2\,T_2^5}-\frac{1}{T_1\,T_2^5}-\frac{3\,T_1}{T_2^5}-$$
$$\frac{T_1^2}{T_2^5}-\frac{6}{T_2^4}-\frac{3}{T_1^7\,T_2^4}+\frac{1}{T_1^5\,T_2^4}+\frac{6}{T_1^4\,T_2^4}-\frac{1}{T_1^3\,T_2^4}+\frac{1}{T_1\,T_2^4}-\frac{T_1}{T_2^4}+\frac{3\,T_1^3}{T_2^4}-\frac{2}{T_2^3}+\frac{3}{T_1^7\,T_2^3}-\frac{3}{T_1^5\,T_2^3}-\frac{1}{T_1^4\,T_2^3}+\frac{2}{T_1^3\,T_2^3}-$$
$$\frac{5}{T_1^2\,T_2^3}+\frac{5}{T_1\,T_2^3}+\frac{T_1}{T_2^3}+\frac{3\,T_1^2}{T_2^3}-\frac{3\,T_1^4}{T_2^3}+\frac{2}{T_2^2}-\frac{1}{T_1^7\,T_2^2}+\frac{3}{T_1^5\,T_2^2}-\frac{5}{T_1^3\,T_2^2}-\frac{2}{T_1^2\,T_2^2}+\frac{5\,T_1}{T_2^2}-\frac{3\,T_1^3}{T_2^2}+\frac{T_1^5}{T_2^2}-\frac{18}{T_2}-$$
$$\frac{3}{T_1^6\,T_2}-\frac{1}{T_1^5\,T_2}+\frac{1}{T_1^4\,T_2}+\frac{5}{T_1^3\,T_2}+\frac{18}{T_1\,T_2}-\frac{5\,T_1^2}{T_2}-\frac{T_1^3}{T_2}+\frac{T_1^4}{T_2}+\frac{3\,T_1^5}{T_2}+18\,T_2-\frac{3\,T_2}{T_1^5}-\frac{T_2}{T_1^4}+\frac{T_2}{T_1^3}+\frac{5\,T_2}{T_1^2}-$$
$$18\,T_1\,T_2-5\,T_1^3\,T_2-T_1^4\,T_2+T_1^5\,T_2+3\,T_1^6\,T_2-2\,T_2^2-\frac{T_2^2}{T_1^5}+\frac{3\,T_2^2}{T_1^3}-\frac{5\,T_2^2}{T_1}+2\,T_1^2\,T_2^2+5\,T_1^3\,T_2^2-3\,T_1^5\,T_2^2+$$
$$T_1^7\,T_2^2+2\,T_2^3+\frac{3\,T_2^3}{T_1^4}-\frac{3\,T_2^3}{T_1^2}-\frac{T_2^3}{T_1}-5\,T_1\,T_2^3+5\,T_1^2\,T_2^3-2\,T_1^3\,T_2^3+T_1^4\,T_2^3+3\,T_1^5\,T_2^3-3\,T_1^7\,T_2^3+6\,T_2^4-$$
$$\frac{3\,T_2^4}{T_1^3}+\frac{T_2^4}{T_1}-T_1\,T_2^4+T_1^3\,T_2^4-6\,T_1^4\,T_2^4-T_1^5\,T_2^4+3\,T_1^7\,T_2^4-6\,T_2^5+\frac{T_2^5}{T_1^2}+\frac{3\,T_2^5}{T_1}+T_1\,T_2^5-3\,T_1^2\,T_2^5+3\,T_1^3\,T_2^5-$$
$$T_1^4\,T_2^5+6\,T_1^5\,T_2^5-3\,T_1^6\,T_2^5-T_1^7\,T_2^5-2\,T_2^6+3\,T_1\,T_2^6-3\,T_1^5\,T_2^6+2\,T_1^6\,T_2^6+T_1^2\,T_2^7-3\,T_1^3\,T_2^7+3\,T_1^4\,T_2^7-T_1^5\,T_2^7\Big\}$$

pdf

In[•]:=
```
PolyPlot[Θ[Knot[#]], ImageSize → Small] & /@ {"K11n34", "K11n42"}
```

Out[•]=
pdf

{  ,  }

tex

Torus knots have particularly nice-looking $\Theta$ invariants. Here are the torus knots $T_{13/2}$, $T_{17/3}$, $T_{13/5}$, and $T_{7/6}$:

pdf

In[∘]:=
```
GraphicsGrid[{
  TubePlot[TorusKnot @@ #] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}},
  PolyPlot[Θ[TorusKnot @@ #]] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}
 }]
```

Out[∘]=
pdf



tex

The next line shows the computation time is seconds for the 132-crossing torus knot $T_{22/7}$ on a 2024 laptop, without actually showing the output. The output plot is in Figure~\ref{fig:T227}.

pdf

In[∘]:=
```
AbsoluteTiming[Θ[TorusKnot[22, 7]];]
```

Out[∘]=
pdf

```
{715.344, Null}
```

In[∘]:=
```
AbsoluteTiming[Θ[Mirror@TorusKnot[22, 7]];]
```

Out[∘]=

```
{1364.81, Null}
```

tex

\begin{figure}

pdf

In[•]:=
```
ImageCompose[PolyPlot[Θ[TorusKnot[22, 7]], ImageSize → 720],
 TubePlot[TorusKnot[22, 7], ImageSize → 360], {Right, Bottom}, {Right, Bottom}]
```

Out[•]=
pdf



tex

\caption{The 132-crossing torus knot $T_{22/7}$ and a plot of its $\Theta$ invariant} \label{fig:T227}
\end{figure}

In[•]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

In[∘]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 10],
  Spacings → 0, Dividers → All, ImagePadding → None, PlotRangePadding → None]
```

Out[∘]=



See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[∘]:=
```
DunfieldKnots = ReadList["../../People/Dunfield/nmd_random_knots"] /. k_Integer ⧴ k + 1;
DK[n_] := DunfieldKnots⟦n - 2⟧;
```

In[•]:=
```
AbsoluteTiming[th = Θ[DK[50]];]
PolyPlot[th]
```

Out[•]=

{19.038, Null}

Out[•]=



In[•]:=
```
AbsoluteTiming[th = Θ[DK[100]];]
PolyPlot[th]
```

Out[•]=
```
{3681.95, Null}
```

Out[•]=



In[∘]:= PolyPlot$\left[\left\{1, -1+T_1+T_2+T_3+T_1^{-1}+T_2^{-1}+T_3^{-1}\right\} * \Theta[DK[100]]\right]$

Out[∘]=



In[∘]:= PolyPlot$\left[\left\{1, 1+T_1+T_2+T_3+T_1^{-1}+T_2^{-1}+T_3^{-1}\right\} * \Theta[DK[100]]\right]$

Out[∘]=



In[•]:=
```
AbsoluteTiming[th = Θ[DK[101]];]
PolyPlot[th]
```

Out[•]=

{3379.04, Null}

Out[•]=